

## BERKELEY MADONNA CODE - FOR SINGLE DOSE CASE

```
;Single Dose - ICp=Cmax only
METHOD RK4
STARTTIME = 0
STOPTIME=24
DT = 0.05

;Enter dosing parameters
TAU=24           ;inter-dose interval
TABMG=1         ;tablet size available (mg)

;Enter saturable bioavailability parameters
D50=90          ; (mg) will only be used if saturable bioavailability selected
MaxS=0.9        ;maximum saturation of bioavailability

;Enter PK values and PD potency
Ka=0.367        ;1/h
CL=5.0          ;L/h
Vc =46.0        ;L
IC50=0.01       ;mcg/mL
;define the threshold to use for drug holiday calculations (number or formula)
THLD=15*IC50/85

;dosetype - Cmax only

;bioavailability type 1 is F=1 and 2 is saturable
biotype=1

;-----CODE BEGINS-----
;initial value of inhibition
P=10

;calculate ICx based upon IC50
ICP=P*IC50/(100-P)

;Single-Dose PK
Ke=CL/Vc
diff=Ka-Ke

;compute single-dose Tmax
tmax=LOGN(Ka/Ke)/diff

;-----exact dose calculations f=1-----
bio=1
fedose=ICP*Vc/(bio*exp(-Ke*tmax))
;-----exact dose calculations f=1-----

;-----exact dose calculations F=1- MaxS*D/(D50+D)-----
A=(1-MaxS)*exp(-Ke*tmax)
```

```
B=D50*exp(-Ke*tmax)-ICP*Vc
C=-1*ICP*D50*Vc
```

```
sedose=( -1*B + SQRT(B**2-4*A*C))/(2*A)
```

```
;-----exact dose calculations F=1-MaxS*D/(D50+D)-----
```

```
;select the final exact dose based upon bioavailability type
```

```
edose= if (biotype=1) then fedose else sedose
```

```
;compute doses that can be administered
```

```
dose=INT(edose/TABMG)*TABMG
```

```
; ignore available table size for QC
```

```
;dose=edose
```

```
;compute relative bioavailability of administered dose
```

```
F= if (biotype=2) then 1-MaxS*dose/(D50+dose) else 1
```

```
;compute model-predicted Cmax, Ctau, AUC(0-tau), Cavg for administered single-doses
```

```
Co=(Ka*F)/(Vc*diff)
```

```
AF1=(1-exp(-1*Ke*TAU))/Ke
```

```
AF2=(1-exp(-1*Ka*TAU))/Ka
```

```
Cmax=Co*dose*(exp(-1*Ke*tmax)-exp(-1*Ka*tmax));
```

```
Ctau=Co*dose*(exp(-1*Ke*TAU)-exp(-1*Ka*TAU))
```

```
auctau=Co*dose*(AF1-AF2)
```

```
Cavg=AUCtau/TAU
```

```
;compute the concentration and PD effect - time values
```

```
Cp=Co*dose*(exp(-1*Ke*TIME) - exp(-1*Ka*TIME))
```

```
PD=100-100*(Cp/(IC50+Cp))
```

```
;compute the average PD effect
```

```
d/dt (PDAUC) = PD
```

```
init PDAUC = 0
```

```
AvgPD=if TIME >= TAU-0.0001 then PDAUC/TAU else -999
```

```
;compute drug holiday
```

```
HOL = IF (Cp >= THLD) THEN 1 ELSE 0
```


```
d/dt (DHOL) = HOL
```

```
init DHOL = 0
```

```
DrugHol = if TIME >= TAU-0.0001 then (TAU-DHOL) else 0
```

## SLIDER DEFINITIONS

Select: Parameters --> Define Sliders



Biotype: min=1 max=2 increment=1


P: min=10 max=90 increment=10

MaxS: min=0 max=(value < 1) increment=(value < 1) ;note max=1 relative bio=0

Others base upon your data


## BATCH RUNS DEFINITIONS

Select: Parameters --> Batch Runs



Select Parameter P

Fill in # of Runs, Initial Value and Final Value



To execute the batch runs, select parameters --> Repeat Batch Runs

## BERKELEY MADONNA CODE - FOR STEADY-STATE DOSE CASE

```
;Steady-State
METHOD RK4
STARTTIME = 0
STOPTIME=24
DT = 0.05

;Enter dosing parameters
TAU=24           ;inter-dose interval
TABMG=1         ;tablet size available (mg)

;Enter saturable bioavailability parameters
D50=90          ; (mg) will only be used if saturable bioavailability selected
MaxS=0.9        ;maximum saturation of bioavailability

;Enter PK values and PD potency
Ka=0.367        ;1/h
CL=5.0          ;L/h
Vc =46.0        ;L
IC50=0.01       ;mcg/mL
;define the threshold to use for drug holiday calculations (number or formula)
THLD=15*IC50/85

;dosetype 1=ctau, 2=cavg, 3=cmax
dosetype=1

;bioavailability type 1 is F=1 and 2 is saturable
biotype=1

;-----CODE BEGINS-----

;initial value of inhibition
P=10
;calculate ICx based upon IC50
ICP=P*IC50/(100-P)

;Steady-State PK values needed for calculations
Ke=CL/Vc
diff=Ka-Ke
tKedif=1-exp(-1*Ke*TAU)
tKadif=1-exp(-1*Ka*TAU)

;compute steady-state Tmax
tmax=(1/diff)*LOGN(Ka*tKedif/(Ke*tKadif))

;-----begin exact dose calculations f=1-----
bio=1
fracss1=exp(-1*ke*TAU)/tKedif
fracss2=exp(-1*ka*TAU)/tKadif
```

```
fracss3=exp(-1*ke*tmax)/tKedif
fracss4=exp(-1*ka*tmax)/tKadif
```

```
fedosetau=(ICp*Vc*diff)/(Ka*bio*(fracss1-fracss2)) ;exact dose for Ctau=ICp
fedoseavg=ICp*CL*TAU/bio ;exact dose for Cavg=ICp
fedosemax=(ICp*Vc*diff)/(Ka*bio*(fracss3-fracss4)) ;exact dose for Cmax=ICp
```

```
;select the exact dose for the type of interest
fedose = if (dosetype=1) then fedosetau else (if (dosetype=2) then fedoseavg else fedosemax)
;-----end exact dose calculations f=1-----
```

```
;-----begin exact dose calculations F=1 - MaxS*D/(D50+D)-----
```

```
A1dtau=exp(-Ke*TAU)
A2dtau=exp(-Ka*TAU)
A3dtau=1-A1dtau
A4dtau=1-A2dtau
Adtau=Ka*(1-MaxS)*(A1dtau*A4dtau-A2dtau*A3dtau)
Bdtau=Ka*D50*(A1dtau*A4dtau-A2dtau*A3dtau)-ICP*Vc*(Ka-Ke)*A4dtau*A3dtau
Cdtau=-1*ICP*D50*Vc*(Ka-Ke)*A4dtau*A3dtau
```

```
A1dmax=exp(-Ke*tmax)
A2dmax=exp(-Ka*tmax)
A3dmax=exp(-Ke*TAU)
A4dmax=exp(-Ka*TAU)
A5dmax=1-A3dmax ;(1-ketau)
A6dmax=1-A4dmax ;1-katau
Admax=Ka*(1-MaxS)*(A1dmax*A6dmax-A2dmax*A5dmax)
Bdmax=Ka*D50*(A1dmax*A6dmax-A2dmax*A5dmax)-ICP*Vc*(Ka-Ke)*A5dmax*A6dmax
Cdmax=-1*ICP*D50*Vc*(Ka-Ke)*A5dmax*A6dmax
```

```
Adavg=1-MaxS
Bdavg=D50-1*ICP*CL*TAU
Cdavg=-1*ICP*CL*TAU*D50
```

```
sedosetau=( -1*Bdtau + SQRT(Bdtau**2-4*Adtau*Cdtau))/(2*Adtau) ;exact dose for
Ctau=ICp
sedoseavg=( -1*Bdavg + SQRT(Bdavg**2-4*Adavg*Cdavg))/(2*Adavg) ;exact dose for
Cavg=ICp
sedosemax=( -1*Bdmax + SQRT(Bdmax**2-4*Admax*Cdmax))/(2*Admax) ;exact dose for
Cmax=ICp
```

```
;select the saturable edose for the type of interest
sedose = if (dosetype=1) then sedosetau else (if (dosetype=2) then sedoseavg else
sedosemax)
;-----end exact dose calculations F=1 - MaxS*D/(D50+D)-----
```

```
;select the final exact dose based upon bioavailability type
edose= if (biotype=1) then fedose else sedose
```

```
;compute doses that can be administered
dose=INT(edose/TABMG)*TABMG
```

```
;ignore tablet sizes available for QC
dose=edose
```

```
;compute relative bioavailability of administered dose
F= if (biotype=2) then 1-MaxS*dose/(D50+dose) else 1
```

```
-----Compute Steady-State Cmax, Ctau, AUCtau, and Cavg-----
Co=(Ka*F)/(Vc*diff)
cmax=Co*dose*((exp(-1*ke*tmax)/tKedif)-(exp(-Ka*tmax)/tKadif))
ctau=Co*dose*((exp(-1*ke*TAU)/tKedif)-(exp(-1*Ka*TAU)/tKadif))
AUCtau=dose*F/CL
cavg=auctau/TAU
```

```
-----Compute values of Cp and PD versus time -----
; Steady-State 1 cmt oral administration
Cp=Co*dose*(exp(-1*Ke*TIME)/tKedif - exp(-1*Ka*TIME)/tKadif)
;compute the PD-time values: direct inhibitory response model
PD=100-100*(Cp/(IC50+Cp))
```

```
-----Compute the Average PD Effect-----
;area under the PD curve divided by inter-dose interval
d/dt (PDAUC) = PD
init PDAUC = 0
AvgPD=if TIME >= TAU-0.0001 then PDAUC/TAU else -999
```

```
-----Compute the Drug Holiday-----
;If Cp > THLD assign holiday variable (HOL) a value of 1 else assign a 0
;The area under the HOL curve gives duration that Cp was above THLD. Holiday will be tau -
the value
HOL = IF (Cp >= THLD) THEN 1 ELSE 0
d/dt (DHOL) = HOL
init DHOL = 0
```

```
DrugHol = if TIME >= TAU-0.0001 then (TAU-DHOL) else 0
```

[See above for defining sliders and batch runs.](#)

For sliders include dosetype with min=1 and max=3 increment=1